
;; Technology Diffusion Model for Industrial Clusters - Version 2.0

;; Description:
;; This model simulates the diffusion of technology between innovator and 
;; imitator enterprise agents through random movement and contact within 
;; an industrial cluster.

;; Agent Types:
;; - Innovator: Enterprises that have adopted and actively spread technology
;; - Imitator: Enterprises that may adopt technology through social learning
;; - Informed: Imitators who have learned about the technology
;; - Adopter: Enterprises that have successfully adopted the technology
;; - Non-adopter: Enterprises that have rejected the technology

breed [imitator imitators]      ;; Potential adopters
breed [informed informeds]      ;; Imitators who know about the technology
breed [innovator innovators]    ;; Technology spreaders
breed [nonadopter nonadopters]  ;; Technology rejectors
breed [adopter adopters]        ;; Successful adopters

;; Global Variables

globals [
  total-imitator      ;; Current count of imitator agents
  total-informed      ;; Current count of informed agents
  total-innovator     ;; Current count of innovator agents
  total-nonadopter    ;; Current count of non-adopter agents
  total-adopter       ;; Current count of adopter agents
]

;; Turtle Variables

turtles-own [
  turn-check          ;; Controls random turning behavior
  wall-turn-check     ;; Controls wall avoidance turning
  processing          ;; Timer for informed-to-innovator transition
  innovativeness      ;; Timer for innovator active period
  adoption-check      ;; Random threshold for adoption decision
  nonadopter-check    ;; Random threshold for non-adoption decision
]

;; Setup Procedures

to setup
  ;; Initialize the model
  clear-turtles
  pop-check
  setup-agents
  update-globals
  do-plots
end

to building-draw
  ;; Use mouse to draw building obstacles (grey patches)
  if mouse-down? [
    ask patch mouse-xcor mouse-ycor [ set pcolor grey ]
  ]
end

;; Agent Initialization

to setup-agents
  ;; Create and initialize all agent breeds
  set-default-shape imitator "person"
  set-default-shape informed "person"
  set-default-shape innovator "person"
  set-default-shape nonadopter "person"
  set-default-shape adopter "person"

  ;; Create imitator agents on black patches
  ask n-of initial-imitator patches with [pcolor = black] [
    sprout-imitator 1 [
      set color blue
    ]
  ]

  ;; Create innovator agents on black patches
  ask n-of initial-innovator patches with [pcolor = black] [
    sprout-innovator 1 [
      set color yellow
      set innovativeness technology-period
    ]
  ]
end

;; Main Simulation Loop

to go
  ;; Run one tick of the simulation
  technology-check
  
  ;; Move all agents in small increments
  repeat 5 [ ask imitator   [ fd 0.2 ] display ]
  repeat 5 [ ask informed   [ fd 0.2 ] display ]
  repeat 5 [ ask innovator  [ fd 0.2 ] display ]
  repeat 5 [ ask nonadopter [ fd 0.2 ] display ]
  
  update-globals
  do-plots
  reset-ticks
end

;; Technology Diffusion Logic

to technology-check
  ;; Main diffusion logic - checks agent interactions and state transitions
  
  ;; IMITATOR: Can be informed by innovators or adopters
  ask imitator [
    if any? other turtles-here with [color = yellow] [ inform ]
    if any? other turtles-here with [color = violet] [ inform ]
    wander
  ]

  ;; INNOVATOR: Spreads technology, ages, and may become adopter or non-adopter
  ask innovator [
    if any? other turtles-here with [color = blue] [ inform ]
    wander
    set innovativeness innovativeness - 1
    if innovativeness = 0 [ nonadopt-or-adopt ]
  ]

  ;; INFORMED: Can be influenced by innovators and counts down to transition
  ask informed [
    if any? other turtles-here with [color = yellow] [ inform ]
    wander
    set processing processing - 1
    if processing = 0 [ get-innovator ]
  ]

  ;; NON-ADOPTER: Simply moves without interaction
  ask nonadopter [ wander ]
end

;; Adoption and Information Transfer

to inform
  ;; Transfer technology knowledge from innovator/adopter to imitator
  set nonadopter-check random 100
  
  ifelse nonadopter-check < nonadoption-probability [
    nonadoption  ;; Agent rejects technology
  ] [
    ;; Agent becomes informed and starts processing timer
    ask imitator-on patch-here [
      set breed informed
      set processing processing-period
    ]
    ask informed-on patch-here [ set color pink ]
  ]
end

to get-innovator
  ;; Convert informed agent into an innovator
  set breed innovator
  set color yellow
  set innovativeness technology-period
end

to adopt
  ;; Convert agent into successful adopter
  set breed adopter
  set color violet
end

to nonadopt-or-adopt
  ;; Decision point: agent either adopts or rejects technology
  set adoption-check random 100
  
  ifelse adoption-check < adoption-probability [
    adopt
  ] [
    nonadoption
  ]
end

to nonadoption
  ;; Convert agent into permanent non-adopter
  set breed nonadopter
  set color green
end

;; Movement Behavior (Wandering with Obstacle Avoidance)

to wander
  ;; Random movement for all agents
  set turn-check random 20
  
  if turn-check > 15 [ right-turn ]
  if turn-check < 5  [ left-turn ]
  
  ;; Avoid obstacles (non-black patches)
  if [pcolor] of patch-ahead 1 != black [ wall ]
end

to wall
  ;; Obstacle avoidance behavior
  set wall-turn-check random 10
  
  if wall-turn-check >= 6 [ wall-right-turn ]
  if wall-turn-check <= 5 [ wall-left-turn ]
end

;; Turning Subroutines
to right-turn
  ;; Gentle right turn for wandering
  rt random-float 10
end

to left-turn
  ;; Gentle left turn for wandering
  lt random-float 10
end

to wall-right-turn
  ;; Sharp right turn for wall avoidance
  rt 170
end

to wall-left-turn
  ;; Sharp left turn for wall avoidance
  lt 170
end

;; Reporting and Visualization

to update-globals
  ;; Update global counters with current agent populations
  set total-imitator   (count imitator)
  set total-informed   (count informed)
  set total-innovator  (count innovator)
  set total-nonadopter (count nonadopter)
  set total-adopter    (count adopter)
end

to do-plots
  ;; Update the diffusion curve plot
  set-current-plot "Diffusion curve"
  
  set-current-plot-pen "imitator"
  plot total-imitator
  
  set-current-plot-pen "informed"
  plot total-informed
  
  set-current-plot-pen "innovator"
  plot total-innovator
  
  set-current-plot-pen "nonadopter"
  plot total-nonadopter
  
  set-current-plot-pen "adopter"
  plot total-adopter
end

;; Validation and Error Checking

to pop-check
  ;; Ensure total agent population does not exceed available patches
  if initial-imitator + initial-innovator > count patches [
    user-message (word "This simulation only has room for " count patches " agents.")
    stop
  ]
end

